# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=firecat
pkgver=6.b5205c8
pkgrel=2
pkgdesc='A penetration testing tool that allows you to punch reverse TCP tunnels out of a compromised network.'
groups=('blackarch' 'blackarch-networking' 'blackarch-tunnel')
arch=('any')
url='https://github.com/BishopFox/firecat'
license=('MIT')
depends=()
makedepends=('git')
source=("git+https://github.com/bishopfox/$pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

build() {
  cd $pkgname

  make
}

package() {
  cd $pkgname

  install -Dm 755 firecat "$pkgdir/usr/bin/$pkgname"
  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README.md \
    BishopFox-Tools-Firecat-Overview_Guide.pdf
}

